

Chapter 25: Export And Import Processing

25.01 INTRODUCTION TO EXPORT AND IMPORT

- 01 "Import" is a filePro processing table command.
It reads in the contents of outside files and puts data into filePro files.
- 02 "Export" is a filePro processing table command.
It sends filePro data to outside files.
- 03 filePro does import and export in five different file formats.
filePro calls them: **1. ASCII**
 2. WORD
 3. WORDPERFECT
 4. MULTI and
 5. DIF
- 04 Delimiters and separators. These are used by the programs that read the exported or imported files to know where each piece of information in the file begins and ends.
There are separators for 1) **Records in the file**, and 2) **Fields within the records**.
There are delimiters for **fields** - both starting and ending each field.
- 05 Terminology. The *foreign file* refers to the file being created by export; or being read by import.
 The *filePro file* refers to the file you are standing in when you Request Output with the export or import command.

25.02 RULES FOR EXPORT AND IMPORT COMMANDS

- 01 The export and import commands are only used in Output Processing. While it can be part of a printed Report Output format, normally it occurs in a Processing Only output table.
- 02 The export or import statement MUST be on a **Then** (Action) line by itself.
- 03 You CANNOT have another action on the same line.
- 04 You CANNOT have a comment on the **Then** line. (But you can on the **If** line)

25.03 FOREIGN FILE FORMATS THAT ARE HANDLED

- 01 ASCII - This produces a pure ascii file, with a variety of field and record formats, dependent on the specified separators and delimiters.

1. By definition, filePro creates the foreign file in ASCII format, with ends of lines delimited by the operating system standards:

- In Unix: newline: nl \n chr("10")
- In Windows/DOS: carriage return/newline cr/nl \n chr("13")chr("10")

Chapter 25: Export And Import Processing

25.03 FOREIGN FILE FORMATS THAT ARE HANDLED (Continued)

01 ASCII (Continued)

2. Syntax with specified delimiters and separators:

```
import ascii alias=/pathname/filename r=n f=n o=n c=n
export ascii alias=/pathname/filename r=n f=n o=n c=n
```

WHERE:

import is the command;
ascii is the argument to the command specifying the format of the foreign file;
alias is the handle assigned to foreign file (with full path);
r specifies the record separator;
f specifies the field separator;
o specifies the opening field delimiter;
c specifies the closing field delimiter; and
n is the character used in the foreign file.

EXAMPLE:

```
3 -----
   If: aa eq "X"          'First record - ask operator for file name
   Then: goto contx
4 -----
   If:
   Then: aa(1,*,g)="X"
5 -----
   If:
   Then: qq(15,allo)
6 -----
   If:
   Then: input popup qq "What would you like to call the file? " default "iv
       x00a.txt"
7 -----
   If:
   Then: qr="/apl/share/"{qq
9 -----
contx If: '*****
      Then: ' A. Export ASCII, comma delimited; 1 line/record
10 -----
      If: 'Export, comma-delimited; 1 line/rec
      Then: export ascii exp=(qr) r=\n f=,
11 -----
      If:
      Then: exp(1)=2; exp(2)=3; exp(3)=4; exp(4)=5
12 -----
      If:
      Then: exp(6)=7; exp(7)=8; exp(8)=9; exp(9)=10; exp(10)=@td
13 -----
      If:
      Then: end
14 -----
@done If:
      Then: cls; msgbox "Your Job Has Completed - "<@RS<"Records Selected"
15 -----
      If:
      Then: close exp; end
16 -----
```

Chapter 25: Export And Import Processing

25.03 FOREIGN FILE FORMATS THAT ARE HANDLED (Continued)

01 ASCII (Continued)

RESULTING FOREIGN FILE FROM THE ABOVE CODE:

```
SS3042B,2,RUNWY,NS,,STAINLESS CLAMP, 5" DIA,,A0011S,,06/06/02
SS3021B,2,APPL,NS,,STAINLESS CLAMP, 6" DIA,,A0099S,,06/06/02
CRO60,1,APPL,23,,COLD ROLLED STEEL,TYPE 105,C10,,06/06/02
CR120,3,10.00,23,,COLD ROLLED STEEL,TYPE 208,,06/06/02
```

3. Syntax with one long string per record (no field separators - fields lengths are retained)

```
import ascii alias=/pathname/filename -x f=n
export ascii alias=/pathname/filename -x f=n
```

WHERE

- x** specifies that each record is a single string and each field is fixed length with no field delimiters.
- f** specifies the field separator.

EXAMPLE: NOTE - Replaces lines 9 through 13 in the previous example.

```
9 -----
contx If: '*****'
      Then: ' C. Export ASCII, fixed length fields, 1 line/record
10 -----
      If:
      Then: export ascii exp=(qr) r=\n -x
11 -----
      If:
      Then: exp(1)=2; exp(2)=3; exp(3)=4; exp(4)=5
12 -----
      If:
      Then: exp(6)=7; exp(7)=8; exp(8)=9; exp(9)=10; exp(10)=@CD
13 -----
      If:
      Then: end
14 -----
```

RESULTING FOREIGN FILE FROM THE ABOVE CODE:

```
SS3042B      2  RUNWYNSSTAINLESS CLAMP, 5" DIA          A0011S      01/19/02
SS3021B      2  APPL NSSTAINLESS CLAMP, 6" DIA          A0099S      01/19/02
CRO60        1  APPL 23COLD ROLLED STEEL          TYPE 105    C10         01/19/02
CR120        3  10.0023COLD ROLLED STEEL          TYPE 208    C10         01/19/02
```

4. Syntax to append an export to a file

```
export ascii -a alias=/pathname/filename r=n f=n o=n c=n
```

WHERE

- a** specifies that export is to be appended to an existing file (If file does not exist, filePro will create it).

Chapter 25: Export And Import Processing

25.03 FOREIGN FILE FORMATS THAT ARE HANDLED *(Continued)*

- 02 **WORD** - WordStar format (Old word processor, now defunct). This is actually an ASCII format with comma-delimited, quote enclosed, fields.

1. Syntax - import word alias=/pathname/filename
 export word alias=/pathname/filename

WHERE

Carriage return/line feed Are provided as record separator:
Commas Are provided as field separators: and
Quotes Are provided as delimiters for alphanumeric, date, and time fields.

EXAMPLE: *NOTE - Replaces lines 9 through 13 in the previous example.*

```

9  -----
contx  If: '*****
      Then: '      D. Export WORD
10 -----
      If:
      Then: export word exp=(qr)
11 -----
      If:
      Then: exp(1)=2; exp(2)=3; exp(3)=4; exp(4)=5
12 -----
      If:
      Then: exp(6)=7; exp(7)=8; exp(8)=9; exp(9)=10
13 -----
      If:
      Then: end
14 -----

```

RESULTING FOREIGN FILE FROM THE ABOVE CODE:

```

"5330-3214","HDWE","PLUMB","","","J-BAR, 2' X 1" X 6"","EXTRUDED ALUMINUM","","",""
"5640-0174","STRT","564","NS","","ANGLE, INSULATION 45-DEG","3" #7-45","","",""
"5670-0024","STRT","567","","","ANGLE, WALL 10' SUSPENDED","CEILING","","",""
"5670-0107","MATL","567","","","ANGLE, 12' WALL WHITE","","","",""

```

- 03 **WORDPERFECT** - Corel WordPerfect **merge file** format (for constructing a mailing list).

1. Syntax - import WordPerfect alias=/pathname/filename
 export WordPerfect alias=/pathname/filename

2. See Section 25.12, this Chapter, for examples of code, and resulting foreign file data.

- 04 **MULTI** - Multiplan SALK file format. While Multiplan is a defunct spreadsheet, its format is read by many current spreadsheets.

1. Syntax - import multi alias=/pathname/filename r=n c=n
 export multi alias=/pathname/filename r=n c=n

WHERE

r specifies the row to start, and to fill by rows;
c specifies the column to start, and to fill by columns; and
n is the row or column number or letter.